COMMONWEALTH OF AUSTRALIA

DEPARTMENT OF NATIONAL DEVELOPMENT

BUREAU OF MINERAL RESOURCES, GEOLOGY AND GEOPHYSICS

RECORD No. 1966/8

THE COMPUTATION OF SUN AND STAR OBSERVATIONS USING A CDC 3600 COMPUTER

by

G.D. LODWICK

The information contained in this report has been obtained by the Department of National Development as part of the policy of the Commonwealth Government to assist in the exploration and development of mineral resources. It may not be published in any form or used in a company prospectus or statement without the permission in writing of the Director, Bureau of Mineral Resources, Geology and Geophysics.

CONTENTS

SUMMARY

Instructions are given for preparing data for the calculation of longitude and azimuth or declination using a Control Data Corporation 3600 computer.

1966/8

1. INTRODUCTION

The computer programs described in this Record have been modified from the originals written for the Ferranti 'Sirius' computer (Parkinson, 1963). The first program, code-named TRUENTH, is designed for the evaluation of both the azimuth of a reference mark and the station longitude using the 'altitude method'. The second program, code-named DECLINAT, is for the evaluation of the azimuth of a reference mark, or more usually the magnetic declination, using the 'hour angle method'. For both programs the latitude of the station must be known.

Both programs work to an accuracy of 0.1 minute of arc.

2. DATA INPUT AND OUTPUT

The following conditions apply for both programs:

(a) Latitude is positive or negative according as the station is in the northern or the southern hemisphere. Similarly a northerly solar declination is positive whereas a southerly one is negative.

(b) Longitude is always considered between 0 and 360 degrees measured east from Greenwich, e.g. East Australia 150°, Mexico 245°.

(c) The Greenwich hour angle (G.H.A.) for the sun or a selected star is calculated from the nautical almanac for the time of observation by summing the G.H.A. of the previous hour and that due to the fraction of the hour, which can be read from the appendix.

For sun observations, however, this latter step has been incorporated in the program so that the insertion of only the observation time and the G.H.A. of the last hour in the appropriate positions are required.

Unfortunately the G.H.A. of the sun and of stars increase at slightly different rates (owing to the difference in length of solar and sidereal days); hence, for star observations it is necessary to calculate the G.H.A. fully, leaving blank the time columns for hours and minutes. If it is necessary to distinguish between observations done at the same station within any hour, this can be incorporated in the eight letter name, for example:

AROPA 1
AROPA 2, etc.

Program TRUENTH

Program TRUENTH involves two computations:
(a) azimuth of reference mark
(b) station longitude

The data required are as follows:
1. station name } for identification only
2. date }
3. time (accurate to one fifth of a second if possible)
4. position indication of sun or star east or west of true meridian
5. declination of sun or star
6. latitude
7. altitude of sun or star
8. circle reading of azimuth mark
9. circle reading of sun or star
10. G.H.A. of sun or star

If either the azimuth or the longitude only is required from an observation, it is acceptable to punch the relevant data only. For azimuth, for example, time and G.H.A. will be blank; the correct value of azimuth is computed but the value for longitude will be based on a G.H.A. of zero. Similarly, for a longitude computation, the circle reading of azimuth mark will be blank.

The position indication of the sun or star is required for a unique solution, to distinguish between the two similar triangles on each side of the vertical plane through the meridian. If the observed sun or star is in the eastern sky this should be indicated by a figure 1 in column 24, if in the western sky by a figure 3 (see Appendix A).

The azimuth angle printed out is measured 0° to 360° from north through east.

Program TRUENTH is shown in Appendix B.

Program DECLINAT

Program DECLINAT is designed to evaluate the azimuth of a selected reference mark; however, by substituting the magnetic meridian circle reading for that of the reference mark the magnetic declination can be calculated directly.

The data required are as follows:

1. station name) for identification only
2. date)
3. time (accurate to one fifth of a second if possible)
4. not required
5. declination of sun or star
6. latitude
7. longitude
8. horizontal circle reading of mark or magnetic meridian
9. horizontal circle reading of sun or star
10. G.H.A. of sun or star

No position indication for sun or star is required, as the computer does this by testing the local hour angle. Column 24 (Appendix A) is left blank.

The azimuth angle is given between -180° and $+180^{\circ}$, positive or negative signs indicating respectively east or west of north. This convention is convenient, as DECLINAT is used predominantly for declination calculations. However, when preparing readings from a declinometer, care is required in preserving the sense of both sun and magnetic circle readings, i.e. if the south end of the magnet has been observed (observer facing magnetic north), the circle reading of the direct sun must be inserted; if the north end of the magnetic is observed (observer facing magnetic south), the sun reading must be that of the sun reflected from behind in a mirror.

If this sense has not been preserved, i.e. if the observation has been made towards magnetic north with the sun reflected from behind or towards magnetic south with the sun direct, then add or subtract 180° from the sun circle reading so that the corrected value remains between 0° and 360°, e.g. 30° becomes 210° and 240° becomes 60°.

Diagramatically,



Program DECLINAT is shown in Appendix C.

3. PREPARATION OF INPUT DATA

Blank data sheets have been produced, as shown in Appendix A. Observation data can be entered on these in the field to facilitate the early punching of data cards on completion of the survey.

The data sheets are divided into 99 columns and are used as follows:

Column	Use
1 to 8	Station name
9 to 15	Date (year, month, day)
16 to 22	Time
23 to 24	N (position of sun or star)
25 to 31	Declination (sign considered)
32 to 38	Latitude (sign considered)
39 to 45	Altitude (for TRUENTH) Longitude (for DECLINAT)
46 to 52	Mark circle reading (for TRUENTH) Magnetic circle reading (for DECLINAT)
53 to 59	Horizontal circle reading of sun or star
60 to 66	Greenwich hour angle

Notes

(a) For the station name, any or all of columns 1 to 8 can be used.
(b) The date must be written as year (abbreviated), month, and day, e.g. 650801 is 1 August 1965. Column 9 is left blank for clarity.
(c) Time is printed in hours, minutes, seconds, and tenths of seconds, but if the G.H.A. is calculated fully and inserted (as is invariably done with star observations), the minutes and seconds columns must be left blank or the computer will increment the G.H.A. by the corresponding amount.
(d) The value for N is inserted in column 24.
(e) All angles are to be given in degrees, minutes, and tenths of minutes, without blanks or decimal points.
(f) For latitude and declination, positive signs are redundant.

Examples of data sheets are given in Appendix A.

4. REFERENCES

PARKINSON, W. D. 1963 Machine computation of sun and star observations. Bur.Min.Resour.Aust.Rec. 1963/39.

APPENDIX A

Examples of data sheets

The following stations have been selected to illustrate the common uses of the programs

Station	Program	Calculation
WA102	DECLINAT	declination (sun)
Sandown 1	TRUENTH	azimuth (sun)
Cookhamdean	TRUENTH	longitude (star)
Aropa 1	TRUENTH	azimuth and longitude (sun)
Sandown 2	DECLINAT	azimuth (sun)

1 2 3 4 5 6 7 8	9 10 11 12 13 14 15	16 17 18 19 20 21 22	23 24	25 26 27 28 29 30 31	32 33 34 35 36 37 38	39 40 41 42 43 44 45	46 47 48 49 50 51 52	53 54 55 56 57 58 59	60 61 62 63 64 65 66
STATION	YRMHDY	HRMNSOS	N	DEC(N+/S-)	LAT(N+/S-)	ALTITUDE OR LONG	AZ MARK MAGNET AZ	AZ SUN	GHA HOUR
WA102	621107	074170		-15520	-33100	116505	274592	191490	289051
SANDOWN 1	520604	1725	3	22292	50388	24040	349335	48371	
COOKHAMD	520408	21	1	19256	51337	35094			303171
AROPA 1	630721	053550	3	20368	-6205	26539	22028	149156	235259
SANDOWN 2	520604	17		22294	50388	1097	349334	51055	83300

Notes. a) For Cookhamdean a star is used; therefore the G.H.A. must be calculated fully before insertion and the minutes and seconds columns left blank. Similarly, although the sun has been used for Sandown 2, the G.H.A. has been calculated fully, so again the minutes and seconds columns are blank.

b) For WA102 and Aropa 1 (both sun observations), the G.H.A. has been inserted for the last hour only and the full time inserted; the machine will incrementthe G.H.A. to the correct value automatically.

c) For Sandown 1 (no longitude required), blanks have been left for G.H.A. but the time has been inserted (for identification only) to the nearest minute. The longitude here calculated is based on a G.H.A. of zero, plus an increment due to 25 minutes. Similarly for Cookhamdean the correct longitude value is calculated but the value for azimuth is that of a mark with the same circle reading as the star (both 0°), i.e. the azimuth of the star.

APPENDIX B

Program TRUENTH

```
      DIMENSION A(6),D(6)
      TYPEREAL LHA,LONG
      WRITE(61,20)
   20 FORMAT(11X,7HSTATION,9X,4HDATE,8X,4HTIME,9X,7HAZIMUTH,3X,
     19H LATITUDE,3X,9HLONGITUDE)
      WRITE(61,21)
   21 FORMAT(26X,7HY   M   D,6X,3HGMT,9X,8HE FROM N,//)
    5 READ(60,1)NAME,IYEAR,MONTH,IDAY,IHOUR,MINUTE,SECS,N,(A(I),I=1,6)
      IF(EOF,60)7,8
    1 FORMAT(A8,I3,I2,I2,I2,I2,F3.1,I2,6F7.3)
    8 DO4 I=1,6
      K=A(I)
      B=A(I)-K
      C=B*5./3.
    4 D(I)=(K+C)*3.14159/180.
      X1=SECS/60.
      X4= (MINUTE+X1)*3.14159/720.
      GHA=D(6)+X4
      AZ=ACOS((SIN(D(1))-SIN(D(2))*SIN(D(3)))/(COS(D(2))*COS(D(3))))
      LHA=ACOS((SIN(D(3))-SIN(D(2))*SIN(D(1)))/(COS(D(2))*COS(D(1))))
      IF(N.GT.2)2,3
    2 AZ=2.*3.14159-AZ
      GOTO15
    3 LHA=2.*3.14159-LHA
   15 AZ=AZ+D(4)-D(5)
      IF(AZ.LT.0.)11,19
   11 AZ=AZ+2.*3.14159
   19 IF(AZ.LT.2.*3.14159)14,10
   10 AZ=AZ-2.*3.14159
   14 AZ=AZ*180./3.14159
      L=AZ
      X=AZ-L
      W=X*60.
      IF(GHA.LT.2.*3.14159)13,17
   17 GHA=GHA-2.*3.14159
   13 LONG=LHA-GHA
      IF(LONG.LT.0.)16,12
   16 LONG=LONG+2.*3.14159
   12 LONG=LONG*180./3.14159
      N1=LONG
      Y1=LONG-N1
      W1=Y1*60.
C     BIAS SECS TO AVOID MINUS SIGN IN PRINT OUT
      SECS = SECS + 0.001
      JYEAR=IYEAR+1900
      NA = A(2)
      ZA=ABSF((A(2)-NA)*100.)
    9 WRITE(61,6)NAME,JYEAR,MONTH,IDAY,IHOUR,MINUTE,SECS,L,W,NA,ZA,N1,W1
    6 FORMAT(11X,A8,I8,I3,I3,I6,I3,F5.1,I7,F5.1,I7,F5.1,I7,F5.1)
      GOTO5
    7 CONTINUE
      END
```

APPENDIX C

Program DECLINAT

```
      DIMENSION A(6),D(6)
      TYPEREAL LHA
      WRITE(61,20)
   20 FORMAT(11X,7HSTATION,9X,4HDATE,8X,4HTIME,6X,11HDECLINATION,3X,
     18HLATITUDE,3X,9HLONGITUDE)
      WRITE(61,21)
   21 FORMAT(26X,7HY   M   D,6X,3HGMT,//)
    5 READ(60,1)NAME,IYEAR,MONTH,IDAY,IHOUR,MINUTE,SECS,N,(A(I),I=1,6)
      IF(EOF,60)7,8
    1 FORMAT(A8,I3,I2,I2,I2,I2,F3.1,I2,6F7.3)
    8 DO4 I=1,6
      K=A(I)
      B=A(I)-K
      C=B*5./3.
    4 D(I)=(K+C)*3.14159/180.
      X1=SECS/60.
      X4= (MINUTE+X1)*3.14159/720.
      GHA=D(6)+X4
      IF(GHA.LT.2.*3.14159)17,16
   16 GHA=GHA-2.*3.14159
   17 HOURANGL=GHA+D(3)
      IF(HOURANGL.LT.2.*3.14159)19,18
   18 HOURANGL=HOURANGL-2*3.14159
   19 IF(HOURANGL.LT.3.14159)21,20
   21 LHA=HOURANGL
      GOTO22
   20 LHA=2.*3.14159-HOURANGL
   22 AZ=ATAN(  SIN(LHA)/(TAN(D(1))*COS(D(2))-SIN(D(2))*COS(LHA))  )
      IF(HOURANGL.LT.3.14159)24,23
   24 AZIMUTH=2.*3.14159-AZ
      GO TO 15
   23 AZIMUTH=AZ
   15 AZIMUTH=AZIMUTH+D(4)-D(5)
      IF(AZIMUTH.LT.0.)11,14
   11 AZIMUTH=AZIMUTH+2.*3.14159
   14 IF(AZIMUTH.LT.2.*3.14159)13,10
   10 AZIMUTH=AZIMUTH-2.*3.14159
   13 AZIMUTH=AZIMUTH*180./3.14159
      L=AZIMUTH
      X=AZIMUTH-L
      W=X*60.
      NA = A(2)
      ZA=ABSF((A(2)-NA)*100.)
      NB=  A(3)
      ZB=(A(3)-NB)*100.
C     BIAS SECS TO AVOID MINUS SIGN IN PRINT OUT
      SECS = SECS + 0.001
      JYEAR=IYEAR+1900
      IF(L.GT.179)25,9
   25 L=L-359
      W=60.-W
    9 WRITE(61,6)NAME,JYEAR,MONTH,IDAY,IHOUR,MINUTE,SECS,L,W,NA,ZA,NB,ZB
    6 FORMAT(11X,A8,I8,I3,I3,I6,I3,F5.1,I7,F5.1,I7,F5.1,I7,F5.1)
      GO TO 5
    7 CONTINUE
      END
```